

    
      
          
            
  
ErgodicControl.jl - Design of Ergodic Trajectories

ErgodicControl.jl is a Julia package for performing ergodic control—that is, designing trajectories that are ergodic with respect to some spatial distribution.
Currently, two trajectory generation methods are implemented: projection-based optimization and spectral multi-scale coverage.
The plan is to implement others, including sequential action control, HEDAC, and sampling based methods.

As of April 2018, this code is under active development and the interface may change.

Installation

It is easy to install. Simply open up Julia and run the following:

Pkg.clone("https://github.com/dressel/ErgodicControl.jl")





Julia 0.6 and the StatsBase package are currently required.

To install the plotting package, run the following:

Pkg.clone("https://github.com/dressel/ErgodicControlPlots.jl")









          

      

      

    

  

    
      
          
            
  
Ergodic Control Overview

Ergodic control is concerned with the generation of ergodic trajectories.
This section defines ergodicity and shows how many of these trajectories are generated.


Ergodicity

A trajectory is ergodic with respect to a distribution if its time-averaged statistics match the distribution’s spatial statistics.
In an ergodic trajectory, the time spent in a region is proportional to the distribution’s density in the region.
If we have some domain \(X\), we denote the spatial distribution \(\phi\), and \(\phi(x)\) describes the density at a point \(x\in X\).

A commonly used metric for ergodicity uses Fourier coefficients [1].
The spatial distribution is decomposed into Fourier coefficients \(\phi_k\):


\[\phi_k(x) = \int_X \phi(x) F_k(x) dx,\]

where \(k\) is a wave-number vector with dimensionality equal to the spatial domain’s.
That is, if the spatial domain has two dimensions, then \(k\) is a vector of length 2.

The function \(F_k(x)\) is as follows:


\[F_k(x) = \frac{1}{h_k}\prod_{i=1}^n \cos \left(\frac{k_i\pi}{L_i} x_i\right),\]

where \(h_k\) is a normalizing factor and \(L_i\) is the length of dimension \(i\).


\[h_k = \left(\int_0^{L_1} \int_0^{L_2} \cos^2(\frac{k_1\pi x_1}{L_1}) \cos^2(\frac{k_2\pi x_2}{L_2})dx_1 dx_2 \right)^{1/2}\]

If we have a trajectory \(x(t)\), we can find the Fourier coefficients \(c_k\) of the trajectory:


\[c_k = \frac{1}{T}\int_0^T F_k(x(t))dt,\]

The ergodic metric \(\mathcal{E}\) is a measure of the difference between trajectory and distribution coefficients:


\[\mathcal{E} = \sum_k \Lambda_k | c_k - \phi_k |^2,\]

where \(\Lambda_k\) is weighted to favor low-frequency features. It takes the form


\[\Lambda_k = \frac{1}{\left(1 + ||k||^2\right)^{(n+1)/2}},\]

where \(n\) is the number state variables in our distribution; for distributions over \(\mathbb{R}^2,\ n = 2\).




Ergodicity in SE(2)

A robot in SE(2) has a state \(x\in\mathbb{R}^3\), where \(x\) consists of the robot’s \(x\)-location, \(y\)-location, and heading \(\theta\). We might denote this state \(x(t) = [x_r, y_r, \theta]\).

The basis functions are as follows:


\[F_{m,n,p}(x) = i^{n-m}\exp\left( i\left[m\psi + (n-m)\theta\right]\right) J_{m-n}(pr),\]

where \((m,n,p)\) are the indices along each direction, \(J_{m-n}\) is the \(m-n\text{th}\) order Bessel function and \((r, \psi, \theta)\) are the robot’s state expressed in polar coordinates:


\[r = \sqrt{x_r^2 + y_r^2},\]


\[\psi = \arctan(y_r / x_r),\]


\[\theta = \theta.\]

The ergodic metric is then expressed


\[\mathcal{E} = \sum_{m,n,p=0}^{M,N,P} \Lambda_{m,n,p} || c_{m,n,p} - \phi_{m,n,p} ||^2,\]

where \(\Lambda_{m,n,p}\) is equivalent to \(\Lambda_k\) where the vector \(k=[m,n,p]\).




Trajectory Generation

We want to generate trajectories with low ergodic score, because an ergodic score of zero implies perfect ergodicity.




Bibliography

[1] G. Mathew and I. Mezic, “Metrics for ergodicity and design of ergodic dynamics for multi-agent systems”







          

      

      

    

  

    
      
          
            
  
Basic Types


Type Aliases

Type aliases are used to simplify tedious code. For example, the type corresponding to a vector of vector of floats is Vector{Vector{Float64}}. To make function arguments easier to read, this is type aliased to the simpler VVF. Below is a list of type aliases used throughout the code and documentation.

typealias MF  Matrix{Float64}
typealias VMF Vector{MF}
typealias VF  Vector{Float64}
typealias VVF Vector{VF}








Trajectory

A discrete trajectory is simply a set of states. Each state is represented a a vector. For example, a point in 2D space is represented with a vector of length 2.

A trajectory is a vector of state vectors (so VVF according to the aliases above).
This might seem tedious, and a simpler representation might have been a 2D array, where each row represents a single state.
The reasoning behind the VVF representation is that slicing vectors out of a 2D array slows Julia code down.
I’m not sure if it has a noticeable effect (if any), but that was the thought process.

To convert between these representations, the traj2mat and mat2traj functions have been included.




Domain

The Domain type has the following fields:

# user provided
mins::Vector{Float64}           # minimum value per dimension
maxes::Vector{Float64}          # maximum value per dimension
cells::Vector{Int}              # number of cells per dimension

# calculated and used internally
lengths::Vector{Float64}        # maxes - mins
cell_lengths::Vector{Float64}   # length of cell in each dimension
num_dims::Int                   # number of dimensions
cell_size::Float64              # size (area, volume, etc) of a cell





The Domain constructor requires knowledge of the domain minimums, maximums, and number of cells per side.

Domain(mins, maxes, cells)





If you wanted to create the unit square with each dimension discretized into 100 bins (100^2 for the whole square), you have the following options:

# verbose
d = Domain([0,0], [1,1], [100,100])

# if one discretization level is provided, all dimensions assume it
d = Domain([0,0], [1,1], 100)

# if you don't provide minimums, they are assumed to be zero
d = Domain([1,1], [100,100])
d = Domain([1,1], 100)

# Unit square in R^2 with 100 cells per side
d = Domain(100)





To generate domains in SE(2), just ensure there are three dimensions and that the last one covers the entire angular space.

d = Domain([-1,-1,-pi], [1,1,pi], 50)





When computing ergodic trajectories over SE(2), it is recommended that you use 50 (or fewer) cells per dimension because SE(2) trajectories use Julia’s besselj function, which makes computation slow.




Probability Distributions

Probability distributions over \(\mathbb{R}^n\) are simply represented as arrays with \(n\) dimensions.

The provided gaussian functions make it easy to generate distributions over a domain

d = Domain([2,1], [200,100])

# returns array with 100 rows and 200 columns
phi = gaussian(d, [1.5,0.5], 0.03*eye(2))

# plot
plot(d, phi)
xlabel("x")
ylabel("y")





The resulting plot is shown below. To learn more about plotting, check out the Visuals section of this readme.

[image: _images/single_domain.png]
The gaussian function can also do multi-Gaussians. In this case the function argument is gaussian(domain, means, Sigmas, weights). The means argument is a vector of mean vectors; Sigmas is a vector of covariance matrices; weights is a vector of weights, one for each Gaussian. This argument is optional and the default value weights matrices equally. The gaussian function will ensure the resulting distribution is a proper density, so you the weights don’t need to add to 1.

d = Domain([1,1], 100)

means = [[.3,.7], [.7,.3]]
Sigmas = [.025*eye(2), .025*eye(2)]
weights = [1,2]
phi = gaussian(d, means, Sigmas, weights)

plot(d, phi)
xlabel("x")
ylabel("y")





[image: _images/multi_domain.png]






          

      

      

    

  

    
      
          
            
  
Ergodic Manager

The abstract ErgodicManager type contains information about the spatial distribution. Currently, there are two sub-types. The type ErgodicManagerR2 manages distributions over \(\mathbb{R}^2\), and the type ErgodicManagerSE2 manages distributions over the special Euclidean group SE(2).


Fields

The ErgodicManagerR2 type has the following fields:

domain::Domain              # spatial domain
K::Int                      # number of Fourier coefficients
phi::Matrix{Float64}        # spatial distribution
phik::Matrix{Float64}       # distribution's Fourier coefficients

# constant regardless of phi (depend on k1,k2)
Lambda::Matrix{Float64}
hk::Matrix{Float64}

# to speed up computation
kpixl::Matrix{Float64}
kpiyl::Matrix{Float64}








Construction

An ergodic manager for \(\mathbb{R}^2\) can be constructed with a Domain, a distribution over that domain, and the number of Fourier coefficients.

ErgodicManagerR2(d::Domain, phi::Matrix{Float64}, K::Int)








Updating Spatial Distribution

The decompose! function decomposes an ergodic manager’s spatial distribution phi into Fourier coefficients, updating the managers phik field:

decompose!(em::ErgodicManager)








Reconstructing Spatial Distributions

Sometimes we want to reconstruct a spatial distribution from the Fourier coefficients, to see how well the Fourier coefficients capture the distribution.

phi = reconstruct(em::ErgodicManager)





If you have your own set of coefficients ck, you can use that instead of em.phik:

phi = reconstruct(em::ErgodicManager, ck::Matrix{Float64})





You could also pass in a trajectory and reconstruct will take care of decomposing it into coefficients and using these to generate a spatial distribution.

phi = reconstruct(em::ErgodicManager, xd::VVF)








Example Managers

I provide a number of pre-made ergodic managers that correspond to frequently used example domains/distributions.

ErgodicManagerR2(example_name::String; K::Int=5, bins::Int=100)





Currently, there are two valid values for example_name: “single gaussian” and “double gaussian”. For exmaple, you could run:

em = ErgodicManagerR2("single gaussian", K=5, bin=100)











          

      

      

    

  

    
      
          
            
  
Trajectory Manager

The TrajectoryManager contains information used during trajectory generation.


Fields and Construction

The TrajectoryManager type has the following fields:

# basic info
N::Int                      # trajectory length
h::Float64                  # time step
x0::Vector{Float64}         # initial state

# cost functions
q::Float64                  # ergodic cost multiplier, default = 1
R::Matrix{Float64}          # control cost multiplier, def = .01*eye(2)
Qn::Matrix{Float64}         # LQ ergodic cost, default = eye(2)
Rn::Matrix{Float64}         # LQ control cost, default = eye(2)
barrier_cost::Float64       # penalizes leaving domain, def = 0

# needed for trajectory generation
initializer::Initializer
descender::Descender        # default is ArmijoLineSearch(10,.1)
dynamics::Dynamics          # default is single integrator





A TrajectoryManager is constructed with basic information about the trajectory and an optional initializer.

TrajectoryManager(x0::Vector{Float64}, h::Float64, N::Int, i::Initializer=RandomInitializer())





By Default,  barrier_cost=0, meaning no barrier cost is applied. When barrier_cost is positive, a quadratic barrier function is added to the objective. This cost penalizes the trajectory for leaving the domain specified in the ErgodicManager during trajectory generation.

The initializer, descender, and dynamics fields are described below.




Initializer

The initializer field must be a subtype of the abstract Initializer type.

A good option is the ConstantInitializer, which just sets every action to the provided value and uses a forward Euler method to determine the trajectory. Below is an example when the control inputs are in \(\mathbb{R}^2\).

tm.initializer = ConstantInitializer([0.0,0.0])








Descender

The descender field must be a subtype of the Descender abstract type.

The default is an ArmijoLineSearch, as Armijo line search has been used extensively in the literature.

Simpler alternatives are shown below. They work ok, but seem to take longer than Armijo line search (which is to be expected). I’ll use them when I’m troubleshooting or if Armijo line search doesn’t work well in a particular problem.

tm.descender = ConstantStep(0.5)        # each step is 0.5
tm.descender = InverseStep(0.5)         # each step is 0.5/iter
tm.descender = InverseRootStep(0.15)    # each step is 0.15/sqrt(iter)








Dynamics

Linear dynamics are common and easy to set up:

tm.dynamics = LinearDynamics(A,B)





The Dubins car is a standard model for cars.

tm.dynamics = DubinsDynamics(v0, r)





If you want to implement your own dynamics, you need to subtype the abstract Dynamics type and implement the linearize and forward_euler functions.

type MyDynamics <: Dynamics
    # fields, constructors, etc

    # following fields must be included
    n::Int      # number of state variables
    m::Int      # number of control variables
end

function linearize(md::MyDynamics, x::VF, u::VF, h::Float64)
    # return A,B matrices
end

function forward_euler(md::MyDynamics, x::VF, u::VF, h::Float64)
    # return new state
end











          

      

      

    

  

    
      
          
            
  
Generating Trajectories

Once you have created an ErgodicManager and a TrajectoryManager, you can generate a trajectory.

xd, ud = pto_trajectory(em, tm)





There are a number of optional keyword arguments. Here they are with their default arguments:

verbose::Bool = true
logging::Bool = false
max_iters::Int = 100
es_crit::Float64 = 0.003
dd_crit::Float64 = 1e-6





When the verbose tag is set to true, progress is presented at each descent iteration. You can turn this to false if you are running trajectory generation as an inner component of a larger algorithm.

The logging tag saves a file temp.csv that contains a copy of each trajectory (just the states, not the actions) at each descent iteration. This is mostly there for gif generation (see Visuals).

The max_iters tag is the maximum number of descent iterations allowed.

The es_crit and dd_crit tags are termination conditions based on the ergodic score and directional derivative, respectively. A lot of research suggests using a directional derivative criterion.

An example using some of these tags is shown below:

pto_trajectory(em, tm, max_iters=1000, dd_crit=1e-4, verbose=false)









          

      

      

    

  

    
      
          
            
  
Visuals


Plotting

To plot the spatial distribution of an ErgodicManager, just use plot.

em = ErgodicManager("single gaussian")
plot(em)





Optional arguments customize the distribution’s appearance. The code alpha argument controls opaqueness; when alpha=1, the distribution is fully opaque and when alpha=0, the distribution is fully transparent. The default value is 1.

The cmap argument controls the colormap used. The default value is "Greys", which uses black for the densest areas and white for the least dense areas. An alternative is the reverse, "Greys_r", which uses white for the densest areas and black elsewhere.

An example using these optional arguments is shown below:

plot(em, cmap="Greys_r", alpha=0.8)





Likewise, you can plot a trajectory too:

plot(em, xd)





In addition to the optional arguments above, there are some additional optional arguments for the trajectory:




GIFs

You can create GIFs of the trajectory generation process. This allows you to see the trajectory at the end of each iteration as it is iteratively improved into the final version. To make such a GIF, you must enable logging when generating the trajectory. You can then call the gif function on the ErgodicManager and TrajectoryManager used. The optional keyword argument fps controls the frames per second. I’ve found 17 to be enjoyable but you can change it if you want.

xd, ud = pto_trajectory(em, tm, max_iters=100, logging=true)
gif(em, tm, fps=17)





You can also make a GIF of the final trajectory, which shows the agent moving along the trajectory. In most cases, this is not particularly interesting, as a simple line can capture the trajectory. However, it is more intersting if the spatial distribution evolves with time (see examples). The function call requires only the ErgodicManager and the trajectory; the optional logging keyword is not needed while generating the trajectory. You can also set the optional keyword argument fps here. It is defaulted to 5.

gif(em, xd)











          

      

      

    

  

    
      
          
            
  
Examples


Single Integrator

using ErgodicControl

em = ErgodicManagerR2("single gaussian", K=5, bins=100)

x0 = [0.4,0.1]
N = 40
h = 0.1

tm = TrajectoryManager(x0, h, N, ConstantInitializer([0.0,0.0]))

xd, ud = pto_trajectory(em, tm)

# plotting
using ErgodicControlPlots
plot(em, xd)





[image: _images/example1.png]



Double Integrator

This example needs to be redone.

using ErgodicControl

em = ErgodicManagerR2("double gaussian", K=5, bins=100)

x0 = [0.49,0.01,0.0,0.0]
N = 40
h = 0.5

tm = TrajectoryManager(x0, h, N, ConstantInitializer([0.0,0.0]))

# dynamics stuff
dynamics!(tm, "double integrator")
tm.Qn = eye(4)

tm.descender = ArmijoLineSearch(1,.01)

xd, ud = pto_trajectory(em, tm)





When doing this, the solver gets stuck. We can try another descent engine.

tm.descender = ArmijoLineSearch(1,.01)





[image: _images/double1.png]



Dubins Car

The Dubins car is simple and often used to model cars.

using ErgodicControl

# ergodic manager
em = ErgodicManagerR2("double gaussian", K=5, bins=100)

# trajectory manager
x0 = [0.5,0.01,pi/4]
N = 40
h = 0.1
tm = TrajectoryManager(x0, h, N, ConstantInitializer([0.0000]))

# things needed for dynamics
tm.dynamics = DubinsDynamics(0.3,0.1)
tm.Qn = eye(3)
tm.R = 0.01 * eye(1)
tm.Rn = 1 * eye(1)

# generate trajectory
xd, ud = pto_trajectory(em, tm)

# plotting
using ErgodicControlPlots
plot(em, xd)





Look at how bad this is! The vehicle leaves our domain!

[image: _images/dubins1.png]
As mentioned before, the trajectory leaves the domain because the Fourier basis function is periodic. This makes sense in the context of the Dubins dynamics. Control effort is only expended when changing the vehicle’s heading.

We can overcome this by penalizing states outside the domain, using the barrier cost we mentioned before. This is handled with the trajectory manager’s barrier_cost field, which is set to 0 by default. Let’s try changing the cost to 1. We’d add the following line before the call to pto_trajectory:

tm.barrier_cost = 1





With this modification, trajectory generation reaches the directional derivative criterion after 155 iterations. The trajectory stays within the domain.

[image: _images/dubins2.png]



Distributions in Three Dimensions

An example if we have a distribution over three dimensions. The agent is a single integrator:

using ErgodicControl

# domain, distribution, and ergodic manager
d = Domain([1,1,1], 100)
means = [[.2,.2,.2], [.8,.8,.2], [.5,.5,.8]]
covs = [0.01*eye(3), 0.01*eye(3), .01*eye(3)]
phi = gaussian(d, means, covs)
K = 5
em = ErgodicManagerR3(d, phi, K)

# trajectory params
x0 = [0.49, 0.01, 0.01]
dt = 0.5
N = 80
tm = TrajectoryManager(x0, dt, N, ConstantInitializer([0.0,0.0,0.0]))
dynamics!(tm, SingleIntegrator(3,dt))
tm.descender = ArmijoLineSearch(1,1e-4)

# trajectory generation and plotting
xd,ud = pto_trajectory(em, tm, dd_crit=1e-4, max_iters=1000)
plot(em, xd, show_score=false)





Plotting is a bit trickier, and is not finished for three dimensions. The tough part is plotting the distribution. Ideally, you’d just plot some isosurfaces for the distribution, but Matplotlib wasn’t made to do such things. I could try Mayavi, but that sounds like a pain. In the following image, I used a scatter plot with points sample from the distribution as a rough representation of the distribution.

[image: _images/three.png]



Time-evolving Spatial Distribution

using ErgodicControl

# Generate the distribution
N = 80
dt = 0.5
T = N*dt
d = Domain([1,1], [100,100])
cov = 0.010 * eye(2)
phi = zeros(100,100,N+1)
for i = 1:N+1
    mui = (.7*(i-1)/N + .15) * ones(2)
    phi[:,:,i] = gaussian(d, mui, cov)
end
ErgodicControl.normalize!(phi, d.cell_size / (N+1))

# Now let's create the ergodic manager in R3
K = 5
em = ErgodicManagerR2T(d, phi, K)

# trajectory params
x0 = [0.49, 0.01]
tm = TrajectoryManager(x0, dt, N, ConstantInitializer([0.,0.]))
tm.R = .1*eye(2)

# I call this second Armijo
tm.descender = ArmijoLineSearch(1,1e-4)

# trajectory generation and plotting
mi = 1000
ddc = 1e-5
v = true
xd,ud = pto_trajectory(em, tm, dd_crit=ddc, max_iters=mi, verbose=v)

# generating the gif
using ErgodicControlPlots
gif(em, xd)





[image: _images/time.gif]



Multi-agent Trajectories

using ErgodicControl

# Set up different domains with different discretizations
d = Domain([1,1], 100)
num_agents = 2

# Set up distribution and ergodic manager
K = 5
means = [[.3,.7], [.7,.3]]
Sigmas = [.025*eye(2), .025*eye(2)]
phi = gaussian(d, means, Sigmas)
em = ErgodicManagerR2(d, phi, K)

# Set up first trajectory manager
x0 = [0.49,0.01]
N = 50
h = 0.6
ci = ConstantInitializer([0.0, 0.0])
tm1 = TrajectoryManager(x0, h, N, ci)
dynamics!(tm1, SingleIntegrator(2,h))

# second tm is like the first, but different starting point
tm2 = deepcopy(tm1)
tm2.x0 = [.79,.99]

# array of trajectory managers
vtm = [tm1, tm2]

# Generate the trajectories
ddc = 1e-4
xd, ud = pto_trajectory(em, vtm, dd_crit=ddc)

# plotting
using ErgodicControlPlots
plot(em, xd, vtm)





[image: _images/multi.png]



Multi-agent Trajectory for Time-evolving Distribution

We can generate a multi-agent trajectory for a time-evolving distribution.

using ErgodicControl

# Generate the distribution
N = 80
dt = 0.5
T = N*dt
d = Domain([1,1], [100,100])
cov = 0.020 * eye(2)
phi = zeros(100,100,N+1)
for i = 1:N+1
    mui = (.7*(i-1)/N + .15) * ones(2)
    phi[:,:,i] = gaussian(d, mui, cov)
end
ErgodicControl.normalize!(phi, d.cell_size / (N+1))

# Now let's create the ergodic manager in R2T
K = 5
em = ErgodicManagerR2T(d, phi, K)

# trajectory params
x0 = [0.49, 0.01, 0., 0.]
tm = TrajectoryManager(x0, dt, N, ConstantInitializer([0.,0.]))
tm.R = .01*eye(2)
tm.descender = ArmijoLineSearch(1,1e-4)
dynamics!(tm, DoubleIntegrator(2,dt))

# create a vector of trajectory managers
tm2 = deepcopy(tm)
tm2.x0 = [.3,.9, 0., 0.]
vtm = [tm, tm2]

# trajectory generation and plotting
mi = 1000
ddc = 1e-5
v = true
xd,ud = pto_trajectory(em, vtm, dd_crit=ddc, max_iters=mi, verbose=v)
gif(em, xd, vtm)





The resulting gif is shown below:

[image: _images/multitime.gif]
The following example is also cool. The multi-agent system consists of a Dubins vehcile and a double integrator.

using ErgodicControl

# Generate the distribution
N = 80
dt = 0.5
T = N*dt
d = Domain([1,1], [100,100])
cov = 0.020 * eye(2)
phi = zeros(100,100,N+1)
for i = 1:N+1
    mui = (.7*(i-1)/N + .15) * ones(2)
    phi[:,:,i] = gaussian(d, mui, cov)
end
ErgodicControl.normalize!(phi, d.cell_size / (N+1))

# Now let's create the ergodic manager
K = 5
em = ErgodicManagerR2T(d, phi, K)

# trajectory params
x0 = [0.5, 0.9, 0., 0.]
tm1 = TrajectoryManager(x0, dt, N, ConstantInitializer([0.,0.]))
tm1.R = .01*eye(2)
dynamics!(tm1, DoubleIntegrator(2,dt))
tm1.barrier_cost = 1.

tm2 = deepcopy(tm1)
dynamics!(tm2, DubinsDynamics(.05, .1))
tm2.initializer = ConstantInitializer([0.05])
tm2.x0 = [.1,.1, .0]

vtm = [tm1, tm2]

# trajectory generation and plotting
xd,ud = pto_trajectory(em, vtm, dd_crit=1e-5, max_iters=1000)
gif(em, xd, vtm)





The resulting gif is shown below

[image: _images/dubins_doubleintegrator.gif]



Distribution over SE(2)

